```
*=============FULL_WAVE_RECTIFIER_POWER============
How energy efficient are full wave rectifiers?
```

```
=======================================================================
FULL_WAVE_RECTIFIER
*
*         _______________________
*        |            VACH      |   |
*       _|_  D1  VAC            |   |
*        ^           _/\  /\  /\_|  _|_
*       / \      |      \/  \/      \ /
*        |       |     RAC      D3  _v_      RDC
*   _____|      _|_                  |   VDC
*  |           / \                   |______/\  /\  /\_
*  |          // \ \  VAC           _|_      \/  \/    |
* ///         \   \//               ___  CDC           |
*              \___/                 |_________________|
*        \ /                        _|_        _|_
*       _v_  D2                D4   / \         ///
*        |          VACL              |
*        |_______|____________________|
*
*
**
.OPTIONS   GMIN=1p        METHOD=trap    ABSTOL=1u      TEMP=27    srcsteps = 1   gminsteps = 1
.OPTIONS   RELTOL=.001    ABSTOL=1n      VNTOL=1u       ITL4=500   ITL1=400
*=========Create_Signal==================
VAC        VAC     VACL     DC      0       SIN(    0     24    60        )
D1         0       VACH     DD
D2         0       VACL     DD
D3         VACH    VDC      DD
D4         VACL    VDC      DD
RAC        VACH    VAC      1m
RDC        VDC     0        10k
CDC        VDC     0        30u
*=========The_Model_Files==================================
.model     DD               D(      IS=3.15e-18 rs = 1m)
.control
*TRAN      TSTEP   TSTOP   TSTART  TMAX     ?UIC?
tran       .1m     120m    20m     .1m
```

```
=======================================================================
Create a 60Hz sine wave with a 24V peak magnitude and
here are the following output waves.
```

```
=======================================================================
set        pensize = 2
plot       vdc vach vacl vac-vacl
```



```
================================================================================
```

Now look at power over time.

```
================================================================================

let         vaci = (vac-vach)/1m
let         VacPWR= vaci*(vac-vacl)
let         rdcPWR=vdc*vdc/10k
plot        VacPWR rdcPWR
```




```
================================================================================
```

Now use some vector processing to find efficiency.
Predict efficiency by subtracting two diodes from the
24volt peak AC waveform.

```
================================================================================
let         VacPWRave  = mean(VacPWR )
let         rdcPWRave  = mean( rdcPWR)
let         lossPWR    = VacPWRave -rdcPWRave
let         EffPWR     = rdcPWRave/VacPWRave
let         DlossRatio = (24-1.6)/24
echo        "VacPWRave =$&VacPWRave rdcPWRave =$&rdcPWRave lossPWR =$&lossPWR  "
echo        "EffPWR =$&EffPWR DlossRatio =$&DlossRatio

================================================================================

VacPWRave =0.0513707 rdcPWRave =0.0479852 lossPWR =0.00338558
EffPWR =0.934095     DlossRatio =0.933333

================================================================================
```

The predicted efficiency comes close to the actual efficiency.

===========Full_Netlist_For_Copy_Paste========================

```
FULL_WAVE_RECTIFIER
*         _______________________
*        |           VACH    |   |
*       _|_ D1 VAC           |  _|_
*        ^       _/\  /\  /\_|  \ /
*       /_\     |   \/  \/      _v_        RDC
*   _____|     _|_    RAC    D3  |   VDC
*  |     |    / _ \              |_______/\  /\  /\_
* _|_    |   // \  \ VAC         |   _|_   \/  \/   |
* ///    |   \   \//             |   ___ CDC        |
*       _|_   \___/             _|_   |_____________|
*       \ /     |                ^        _|_
*       _v_ D2  |            D4 /_\       ///
*        |      |  VACL          |
*        |______|________________|
*

**
.OPTIONS  GMIN=1p       METHOD=trap   ABSTOL=1u     TEMP=27   srcsteps = 1  gminsteps = 1
.OPTIONS  RELTOL=.001   ABSTOL=1n     VNTOL=1u      ITL4=500  ITL1=400
*=========Create_Signal=================
VAC       VAC    VACL    DC     0      SIN(   0    24   60      )
D1        0      VACH    DD
D2        0      VACL    DD
D3        VACH   VDC     DD
D4        VACL   VDC     DD
RAC       VACH   VAC     1m
RDC       VDC    0       10k
CDC       VDC    0       30u
*=========The_Model_Files=================================
.model    DD                D(     IS=3.15e-18 rs = 1m)

.control
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      .1m    120m    20m     .1m
set       pensize = 2
plot      vdc vach vacl vac-vacl

let       vaci = (vac-vach)/1m
let       VacPWR= vaci*(vac-vacl)
let       rdcPWR=vdc*vdc/10k
plot      VacPWR rdcPWR

let       VacPWRave = mean(VacPWR )
let       rdcPWRave    = mean( rdcPWR)
let       lossPWR = VacPWRave -rdcPWRave
let       EffPWR= rdcPWRave/VacPWRave
let       DlossRatio = (24-1.6)/24
echo      "VacPWRave =$&VacPWRave rdcPWRave =$&rdcPWRave lossPWR =$&lossPWR  "
echo      "EffPWR =$&EffPWR DlossRatio =$&DlossRatio   "
.endc
.end
```

9.16.10_12.48PM

dsauersanjose@aol.com
Don Sauer